Tangentes communes à deux paraboles

Fiche professeur

Figure



Correction avec Maxima

1. On définit les fonctions f et g.

(%i1) `f(x):=x^2-1;`

(%o1) $$f(x) := x^2 - 1$$

(%i2) `g(x):=-x^2+2*x-3;`

(%o2) $$g(x) := -x^2 + 2x - 3$$

2. On détermine l'équation réduite de la tangente à $\mathscr{C}_f$ au point A d'abscisse a.

(%i25) `taylor(f(x), x, a, 1);`

(%o25) $$-1 + a^2 + 2a(x-a) + \dots$$

(%i26) `ratsimp(%);`

(%o26) $$2ax - a^2 - 1$$

L'équation réduite de la tangente T_a à $\mathscr{C}_f$ est $y = 2ax - a^2 - 1$.

3. On détermine l'équation réduite de la tangente à $\mathscr{C}_g$ au point B d'abscisse b.

(%i27) `taylor(g(x), x, b, 1);`

(%o27) $$-3 + 2b - b^2 + (-2b+2)(x-b) + \dots$$

(%i28) `ratsimp(%);`

(%o28) $$(2-2b)x + b^2 - 3$$

L'équation réduite de la tangente T_b à $\mathscr{C}_g$ est $y = (2-2b)x + b^2 - 3$.

4. Les deux droites T_a et T_b sont confondues si et seulement si leurs coefficients directeurs sont égaux et leurs ordonnées à l'origine également ; on résout donc le système.

(%i29) `solve([2*a=2-2*b,-a^2-1=b^2-3],[a,b]);`

(%o29) $$[[a = -\frac{\sqrt{3}-1}{2}, b = \frac{\sqrt{3}+1}{2}], [a = \frac{\sqrt{3}+1}{2}, b = -\frac{\sqrt{3}-1}{2}]]$$

On trouve qu'il existe deux tangentes communes, l'une correspondant à $a = -\frac{\sqrt{3}-1}{2}$ et $b = \frac{\sqrt{3}+1}{2}$ et l'autre correspondant à $a = \frac{\sqrt{3}+1}{2}$ et $b = -\frac{\sqrt{3}-1}{2}$.

5. On a les abscisses des points de contact entre les courbes et les tangentes communes, on cherche leurs ordonnées.

(%i30) `f(-(sqrt(3)-1)/2);`

(%o30) $$\frac{\left(1-\sqrt{3}\right)^2}{4} - 1$$

(%i31) `ratsimp(%);`

(%o31) $$-\frac{\sqrt{3}}{2}$$

Un des points de $\mathscr{C}_f$ d'intersection de la courbe avec la tangente est $A_1\left(-\frac{\sqrt{3}-1}{2}\,;-\frac{\sqrt{3}}{2}\right)$.

On cherche l'ordonnée du second point sur $\mathscr{C}_f$.

```
(%i32) f((sqrt(3)+1)/2);
```

$$(\%o32) \qquad \frac{\left(\sqrt{3}+1\right)^2}{4}-1$$

```
(%i33) ratsimp(%);
```

$$(\%o33) \qquad \frac{\sqrt{3}}{2}$$

L'autre point de $\mathscr{C}_f$ d'intersection de cette courbe avec une tangente est $A_2\left(\frac{\sqrt{3}+1}{2}\,;\frac{\sqrt{3}}{2}\right)$.

6. On fait de même avec les points de contact des tangentes communes avec $\mathscr{C}_g$.

```
(%i34) g((sqrt(3)+1)/2);
```

$$(\%o34) \qquad -\frac{\left(\sqrt{3}+1\right)^2}{4}+\sqrt{3}-2$$

```
(%i35) ratsimp(%);
```

$$(\%o35) \qquad \frac{\sqrt{3}-6}{2}$$

```
(%i36) g(-(sqrt(3)-1)/2);
```

$$(\%o36) \qquad -\sqrt{3}-\frac{\left(1-\sqrt{3}\right)^2}{4}-2$$

```
(%i37) ratsimp(%);
```

$$(\%o37) \qquad -\frac{\sqrt{3}+6}{2}$$

Les deux points d'intersection des tangentes avec $\mathscr{C}_g$ sont donc $B_1\left(\frac{\sqrt{3}+1}{2}\,;\frac{\sqrt{3}-6}{2}\right)$ et $B_2\left(-\frac{\sqrt{3}-1}{2}\,;-\frac{\sqrt{3}+6}{2}\right)$.

7. Il reste à calculer les équations réduites des tangentes ; pour cela on remplace dans l'équation de T_a le nombre a par l'abscisse de A_1 puis par l'abscisse de A_2.

```
(%i39) subst(-(sqrt(3)-1)/2, a, %o26);
```

$$(\%o39) \qquad \left(1-\sqrt{3}\right)x-\frac{\left(1-\sqrt{3}\right)^2}{4}-1$$

```
(%i41) expand(%);
```

$$(\%o41) \qquad -\sqrt{3}\,x+x+\frac{\sqrt{3}}{2}-2$$

La tangente commune passant par A_1 a donc pour équation réduite : $y=\left(1-\sqrt{3}\right)x+\frac{\sqrt{3}}{2}-2$.

```
(%i42) subst((sqrt(3)+1)/2, a, %o26);
```

$$(\%o42) \qquad \left(\sqrt{3}+1\right)x-\frac{\left(\sqrt{3}+1\right)^2}{4}-1$$

```
(%i43) expand(%);
```

$$(\%o43) \qquad \sqrt{3}\,x+x-\frac{\sqrt{3}}{2}-2$$

La tangente commune passant par A_2 a donc pour équation réduite : $y=\left(1+\sqrt{3}\right)x-\frac{\sqrt{3}}{2}-2$.

8. Enfin on peut vérifier qu'en remplaçant b par les abscisses de B_1 et de B_2 dans l'équation de T_b, on trouve les mêmes équations de droites.

```
(%i44) subst((sqrt(3)+1)/2, b, %o28);
```

(%o44) $$\left(1-\sqrt{3}\right)\,x+\frac{\left(\sqrt{3}+1\right)^2}{4}-3$$

```
(%i45) expand(%);
```

(%o45) $$-\sqrt{3}\,x+x+\frac{\sqrt{3}}{2}-2$$

```
(%i46) subst((-sqrt(3)+1)/2, b, %o28);
```

(%o46) $$\left(\sqrt{3}+1\right)\,x+\frac{\left(1-\sqrt{3}\right)^2}{4}-3$$

```
(%i47) expand(%);
```

(%o47) $$\sqrt{3}\,x+x-\frac{\sqrt{3}}{2}-2$$